Appendix A: Palindrome filter

In this appendix we illustrate the CLiP style of literate programming by a program to filter palindromic lines from an input file. Three files are involved: the program module (PALIN- DROME.PAS), a test file (TESTDATA.TXT) and a command file (PALINDROME.COM) to run the palindrome filter.

Subsections A.1 through A.7 display the documentation literally, but we have numbered the code lines in the margin for easy reference from the main text.

A.1. Specification

A palindrome is a sentence with the property that the letters from left to right, read the same as from right to left. In the comparison uppercase and lowercase letters are considered to be equivalent and all other characters are simply ignored. Hence an empty sentence is a palindrome. Other examples are:

1 (*************  #file "TESTDATA.TXT"  #comment off  *************)
2 Ada
3 1234567
4 Able was I, ere I saw Elba.
5 A man, a plan, a canal, Panama.
6 Norma is as selfless as I Am, Ron.
7 (*****************  End of TESTDATA.PAS  ************************)

The program PALINDROME reads an input file, filters the lines that are palindromic and writes them to an output file.

A.2. Communication with the outside world

The program conforms to the general template of a Pascal program. We introduce the files IN_FILE and OUT_FILE to define its communication with the outside world. The actual files have to be specified at run-time. Thus we have

 8 (*****************  #file "PALINDROME.PAS"  *********************)
 9 (****************************************************************)
10 (* Program: Palindrome filter program.                          *)
11 (* Purpose: To filter the palindromic lines from a given input  *)
12 (*          file to a specified output file.                    *)
13 (****************************************************************)
14 PROGRAM PALINDROME (INPUT, OUTPUT, IN_FILE, OUT_FILE);

15

16 (*******    Palindrome constants  #multiple #comment off  *******)
17 (*******    Palindrome types      #multiple #comment off  *******)
18 VAR
19     IN_FILE, OUT_FILE: TEXT;
20 (*******    Palindrome variables  #multiple #comment off  *******)

21

22 BEGIN
23     RESET   (IN_FILE);
24     REWRITE (OUT_FILE);

25

26     (*****************  Palindrome (body)  **********************)
27     (** Copy the lines of the IN_FILE that are palindromic to  **)
28     (** the OUT_FILE.                                          **)
29     (************************************************************)

30

31 END (*PALINDROME*).
32 (*******************  End of PALINDROME.PAS  ********************)

To prepare the module for future declarations of constants and types we have

33 (*******        Palindrome constants      #leader #quick  *******)
34 CONST
35 (*******        Palindrome types          #leader #quick  *******)
36 TYPE

A.3. Processing of the files

The program processes IN_FILE line by line. The idea is to buffer an exact copy of the current line in IN_LINE, while at the same time its letters are buffered in LETTERS. So LETTERS will be empty if the line holds no letters at all, in which case the line is considered to be palindromic by definition.

We choose the buffers IN_LINE and LETTERS to be of the same type, TEXT_LINE, which we will not specify in detail right now. For this purpose we introduce a type ABSTRACT.

37 (*******        Palindrome types                  #quick  *******)
38     ABSTRACT  = (DEFINED, UNDEFINED);

TEXT_LINE will temporarily be declared ABSTRACT and its details will be defined later. Thus the declaration of TEXT_LINE

39 (*******                Palindrome types                  *******)
40 (*******            Declaration of TEXT_LINE              *******)
41 (*****************  End of Palindrome types  ********************)

is temporarily satisfied with the type ABSTRACT.

42 (*******    Declaration of TEXT_LINE     #quick #default  *******)
43     TEXT_LINE = ABSTRACT;

The declaration for the variables IN_LINE and LETTERS becomes

44 (*******    Palindrome variables                  #quick  *******)
45     IN_LINE,
46     LETTERS:        TEXT_LINE;

We have to test LETTERS in order to decide whether or not IN_LINE contains a palindrome. The result of this test is flagged by IS_PALINDROME, for which we introduce the declaration

47 (*******    Palindrome variables                  #quick  *******)
48     IS_PALINDROME:  BOOLEAN;

Now the body of the Palindrome filter may be expanded as

49     (*****************  Palindrome (body)  **********************)
50     (** Copy the lines of the IN_FILE that are palindromic to  **)
51     (** the OUT_FILE.                                          **)
52     WHILE NOT EOF (IN_FILE) DO
53     BEGIN
54         (*****************  Palindrome (1)  *********************)
55         (** Read a line from IN_FILE into IN_LINE. The letters **)
56         (** of this line are copied to LETTERS.                **)
57         (********************************************************)

58 59 READLN (IN_FILE); 60

61         (*****************  Palindrome (2)  *********************)
62         (** Test palindromicity of LETTERS. Set IS_PALINDROME  **)
63         (** to reflect the result of the test.                 **)
64         (********************************************************)

65

66         IF IS_PALINDROME THEN
67         BEGIN
68             (*****************  Palindrome (3)  *****************)
69             (** Write IN_LINE to OUT_FILE.                     **)
70             (****************************************************)

71

72             WRITELN (OUT_FILE);
73         END (*IF*);
74     END (*WHILE*);
75     (*************  End of Palindrome (body)  *******************)

A.4. Choosing the structure of IN_LINE and LETTERS

Before we can proceed we need to establish a structure for the objects IN_LINE and LETTERS. Thus we define TEXT_LINE as a structure with two components. The first component is an array, CHARS, which contains the characters to be buffered. The second component, LENGTH, indicates which part of the array is actually occupied. The maximum number of characters that can be buffered by the structure is determined by the length, MAX_L, of the array. MAX_L serves as an implementation parameter.

76 (*******    Palindrome constants                  #quick  *******)
77     MAX_L = 132;

78

79 (*******        Declaration of TEXT_LINE          #quick  *******)
80     TEXT_LINE =     RECORD
81                         CHARS: ARRAY[1..MAX_L] OF CHAR;
82                         LENGTH: 0..MAX_L;
83                     END (*RECORD*);

A.5. Reading a line

For efficiency reasons we fill IN_LINE and LETTERS simultaneously. Therefore we buffer every character that is read from IN_FILE in the variable IN_CHAR.

84 (*******    Palindrome variables                  #quick  *******)
85     IN_CHAR:    CHAR;

Only when IN_CHAR turns out to be a letter it is copied to LETTERS. Since this process is crucial for the overall operation, we make provisions for some debugging code here.

 86         (*****************  Palindrome (1)  *********************)
 87         (** Read a line from IN_FILE into IN_LINE. The letters **)
 88         (** of this line are copied to LETTERS.                **)
 89         IN_LINE.LENGTH := 0;
 90         LETTERS.LENGTH := 0;
 91         WITH IN_LINE DO
 92         WHILE NOT EOLN (IN_FILE) DO
 93         BEGIN
 94             READ (IN_FILE, IN_CHAR);
 95             LENGTH := LENGTH + 1;
 96             CHARS[LENGTH] := IN_CHAR;
 97             IF IN_CHAR IN [’A’..’Z’, ’a’..’z’] THEN
 98             WITH LETTERS DO
 99             BEGIN
100                 LENGTH := LENGTH + 1;
101                 CHARS[LENGTH] := IN_CHAR;
102             END (*WITH/IF*);
103         END (*WHILE/WITH*);

104

105         (*********************  Palindrome (test)  **************)
106         (** Check contents of IN_LINE and LETTERS.  #optional  **)
107         (********************************************************)

108 109 (***************** End of Palindrome (1) **************)

A.6. Testing for palindromicity

We test the palindromicity of LETTERS in two steps. First we transform the contents of LETTERS to uppercase and then we compare the characters of LETTERS pairwise. The comparison is done starting with the most outside characters and progressing inward. The string is assumed a palindrome until the opposite is proven through a pair of different characters. With the local counter

110 (*******    Palindrome variables                  #quick  *******)
111     I:        INTEGER;

we keep track of the comparing process. Now Palindrome (2) can be expanded as

112         (*****************  Palindrome (2)  *********************)
113         (** Test palindromicity of LETTERS. Set IS_PALINDROME  **)
114         (** to reflect the result of the test.                 **)
115         WITH LETTERS DO
116         BEGIN
117             (* Transform lowercase to uppercase.                *)
118             FOR I := 1 TO LENGTH DO
119             IF CHARS[I] IN [’a’..’z’]
120             THEN CHARS[I] :=
121                  CHR(ORD(CHARS[I]) - ORD(’a’) + ORD(’A’));

122

123             (* Perform the palindromicity test.                 *)
124             IS_PALINDROME := TRUE;
125             I := 1;
126             WHILE IS_PALINDROME AND (I <= LENGTH DIV 2) DO
127             IF CHARS[I] = CHARS[LENGTH-I+1] THEN
128                 I := I + 1
129             ELSE
130                 IS_PALINDROME := FALSE;
131         END (*WITH*);
132         (*****************  End of Palindrome (2)  **************)

A.7. Writing the palindrome

The only remaining action is to write the contents of IN_LINE. Again we need a local counter

133 (*******    Palindrome variables                  #quick  *******)
134     J:        INTEGER;

The writing proceeds straight forward.

135             (*****************  Palindrome (3)  *****************)
136             (** Write IN_LINE to OUT_FILE.                     **)
137             WITH IN_LINE DO
138             BEGIN
139                 FOR J := 1 TO LENGTH DO
140                     WRITE (OUT_FILE, CHARS[J]);
141             END (*WITH*);
142             (*************  End of Palindrome (3)  **************)

A.8. Running the Palindrome filter

To run the program in a VAX/VMS environment the following command procedure is convenient.

143 (*********  #file "PALINDROME.COM"  #comment off  ***************)
144 $!***************************************************************!
145 $! Run PALINDROME. Input file and output file are parameters.   *!
146 $! Par1: Specification of input file.                           *!
147 $! Par2: Specification of output file.                          *!
148 $!***************************************************************!

149 $DEFINE IN_FILE  ’P1
150 $DEFINE OUT_FILE ’P2
151 RUN PALIND
152 (*****************  End of PALINDROME.COM  **********************)